
Vreau s, coală ca afară

CS 201J: Engineering Software, University of Virginia

I will not lie, cheat or steal
If I am unsure [if something is] lying, cheating or stealing,
I will ask before doing it
I will not look at solutions from last year’s problem sets
even though I know they are on the web
I will do what I can to help my fellow classmates learn

o să ı̂ntreb când nu ı̂nteleg: probabil nu ı̂nt, eleg nici alt, ii
o să discut temele de casă cu colegii (dacă regula permite)
n-am să le dau solut, ia, ci o să-i ajut s-o descopere ei

I will ask for help
o să ı̂ncerc ı̂ntâi singur
dar o să ı̂ntreb colegii sau profesorii ı̂nainte să devin frustrat

I will provide useful feedback
ambele părt, i trebuie să s, tie dacă (nu) merge bine



Logică s, i structuri discrete

Liste
Marius Minea

marius@cs.upt.ro

http://www.cs.upt.ro/˜marius/curs/lsd/

13 octombrie 2015

mailto:marius@cs.upt.ro
http://www.cs.upt.ro/~marius/curs/lsd/


Liste

Listele sunt un mod de bază de a lucra cu colect, ii de elemente.

O listă e o secvent, ă finită, ordonată, de valori de acelas, i tip.

listele sunt finite, dar pot avea lungime oricât de mare
spre deosebire de tuple (tip separat pentru fiecare lungime):

perechi, triplete, n-tuple
nu pot fi infinite (altă not, iune/alt tip: engl. stream)

ordinea elementelor contează: [1; 3; 2] 6= [3; 1; 2]
spre deosebire de mult, imi

accesul la elementele listei e secvent, ial (acces direct doar la primul)
spre deosebire de tablou, cu acces direct la toate elementele



Listele ca tip recursiv

Listele pot fi definite recursiv:
O listă este o listă vidă, sau un element urmat de o listă.

Mai precis, pentru că listele sunt de un anumit tip:
lista vidă este o listă de elemente de tip a (pentru orice tip)
combinat, ia dintre un element de tip a (ca prim element)

s, i o listă de elemente de tip a (restul = coada listei)
e deasemenea o listă de elemente de tip a.

Definit, ia de mai sus se numes, te s, i inductivă, deoarece defines, te
toate listele (de un anumit tip):

pornind de la cea mai simplă (cazul de bază),
s, i prin modul de a construi o listă mai mare dintr-una mai mică

(pasul inductiv).



Lista ca tip de date ı̂n ML

Tipul de date ’a list : parametrizat cu orice tip de element ’a
int list, string list, etc.

Lista e un tip de date polimorf (gr. “mai multe forme”),
mai precis, avem polimorfism parametric (introdus ı̂n ML, 1975)

câte un tip listă pentru fiecare tip de element, dat ca parametru

Listele se scriu ı̂ntre paranteze drepte [ ] cu ; ı̂ntre elemente:
# [1;3;2] (* o lista de intregi *);;
- : int list = [1; 3; 2]
# ["asta";"e";"o";"lista"] (* o lista de siruri *) ;;
- : string list = ["asta"; "e"; "o"; "lista"]



Construirea de liste

Operatorul :: creează o nouă listă dintr-un element (devine capul
noii liste) s, i o listă (care devine coada noii liste). Lista vidă e [] .

ATENT, IE! :: creează o nouă listă, NU modifică lista dată!
:: e asociativ la dreapta (se poate folosi de mai multe ori).
# 2 :: [3;1];;
- : int list = [2; 3; 1]
# 1 :: 2 :: [];;
- : int list = [1; 2]

@ concatenează două liste. Evitat, i, trebuie să parcurgă prima listă (ineficient)
# [2;1] @ [3];;
- : int list = [2; 1; 3]



Potrivirea de tipare (pattern matching)
Un tip definit prin mai multe variante (listă vidă sau cap+coadă)
se prelucrează prin tipare (cu un tipar pentru fiecare variantă).

Potrivirea de tipare ne permite să descompunem un obiect după
structură, identificând (s, i numind) părt, ile componente.

match lista with
| [] -> expr1
| cap :: coada -> expr2

Această construct, ie e o expresie, cu rezultat expr1 dacă lista e vidă;
altfel, identificatorii cap s, i coada sunt legat, i la cele două părt, i ale
listei, s, i pot fi folosit, i ı̂n expr2, care dă rezultatul ı̂ntregii expresii

Mai concis, cu function definim o funct, ie de argument listă cu
rezultatul dat de valoarea expresiei–tipar:

function
| [] -> expr1
| cap :: coada -> expr2

Bara | la prima variantă e opt, ională dar poate face codul mai us, or de citit



Funct, ii predefinite cu liste

Modulul List oferă o serie funct, ii pentru lucrul cu liste.
Le folosim cu numele List.numefunct, ie .
(e preferabil fat, ă de deschiderea moduluilui open List s, i folosirea
numelui simplu, pentru că e mai clar ı̂n cod că se folosesc liste)

Funct, iile cele mai simple: List.hd (head) s, i List.tl (tail)
returnează capul, s, i respectiv coada listei.

# List.hd [1;4;3];;
- : int = 1
# List.tl [1;4;3];;
- : int list = [4; 3]

Alte funct, ii: length, mem (membru), nth (al n-lea element), etc.



Potrivire de tipare sau hd / tl ?

Funct, iile hd s, i tl nu sunt definite pentru lista vidă (sunt funct, ii
part, iale pe tipul listă). La apel cu [] ele generează o except, ie.

# List.hd [];;
Exception: Failure "hd".
# List.tl [];;
Exception: Failure "tl".

Vom discuta ı̂n alt curs despre tratarea except, iilor.

Într-o funct, ie, trebuie să tratăm toate cazurile.

Folosind potrivirea de tipare (cu match ... with sau function)
compilatorul verifică s, i ne asigură că nu am uitat nicio variantă.
Folosind hd s, i tl trebuie să ne asigurăm noi că nu avem lista vidă.

Preferăm de aceea accesul la cap/coadă prin potrivirea de tipare.



Funct, ii simple: lungimea unei liste

let rec len = function
| [] -> 0
| _ :: t -> 1 + len t

Tiparul _ se potrives, te cu orice. Folosim pentru a ignora valoarea.

Variantă: ı̂n parcurgere, acumulăm rezultatul part, ial
⇒ ı̂ncă un parametru: elementele numărate până acum

let rec len2 r = function
| [] -> r
| _ :: t -> len2 (r+1) t

let len lst = len2 0 lst

(init, ial, nu am numărat niciun element, de aici argumentul 0)

Modulul List defines, te funct, ia List.length .



Definit, ii locale

let rec len2 r = function
| [] -> r
| _ :: t -> len2 (r+1) t

let len lst = len2 0 lst

se poate rescrie:

let len lst =
let rec len2 r = function

| [] -> r
| _ :: t -> len2 (r+1) t

in len2 0 lst

Putem citi ı̂n felul următor: let len lst = len2 0 lst
unde definit, ia funct, iei len2 e dată (s, i vizibilă) doar ı̂n interior.

Doar funct, ia len2 e recursivă, nu s, i len (doar foloses, te len2).



Recursivitatea finală (prin revenire; tail recursion)

În varianta
let rec len = function

| [] -> 0
| _ :: t -> 1 + len t

adunarea se face abia la revenirea din apel: pentru o listă de
lungime n vom avea n apeluri active când ajungem la sfârs, itul ei.
⇒ va es, ua prin depăs, irea stivei la liste foarte lungi

Preferăm varianta 2:

let len lst =
let rec len2 r = function

| [] -> r
| _ :: t -> len2 (r+1) t

in len2 0 lst

Apelul recursiv e ultima operat, ie pe acea ramură (tail recursion).
⇒ fiecare apel recursiv returnează aceeas, i valoare ca cel anterior
Nu se fac calcule la revenire, nu trebuie stivă, rezultatul poate fi
returnat direct. Compilatorul va optimiza apelul ı̂n iterat, ie (ciclu).

Recursivitatea finală e practic la fel de eficientă ca iterat, ia.



Funct, ii simple: testul de membru

Apare valoarea x ı̂n listă ?

let rec mem x = function
| [] -> false
| h :: t -> x = h || mem x t

val mem : ’a -> ’a list -> bool = <fun>

x e membru ı̂n listă dacă:
x e capul listei h, SAU
x e membru ı̂n coada listei t

Operatorul ||: SAU logic: cel put, in un operand adevărat (true)
dacă primul e true, rezultatul e true ⇒ nu mai evaluează al doilea

List.mem e deasemenea o funct, ie predefinită .



Tipuri de date algebrice

Limbajele funct, ionale au teoria tipurilor fundamentată matematic,
ele permit definirea de noi tipuri (compuse) după anumite reguli.

Un tip de date algebric e definit ca o combinat, ie (posibil recursivă)
de alte tipuri. Variante comune sunt:

Tipul produs (produsul cartezian A× B a două tipuri A s, i B)
type physunit = float * string
O valoare a acestui tip e o pereche (̂ın general: tuplu) de elemente
din tipurile componente.
Există s, i variante cu nume pentru câmpuri (ca structurile ı̂n C).

Tipul sumă (tip uniune, sau tip cu variante).
În ML, fiecare variantă are un constructor de tip, notat printr-o
etichetă (tag), pentru a deosebi variantele ı̂ntre ele
type mix = Int of int | Real of float | Str of string

Combinând tipuri sumă/produs definim tipul potrivit problemei.



Listele ca tip de date algebric
Tipul listă nu e “special”, ı̂l putem defini ca tip de date algebric:
– un constructor Cons care creează o nouă listă dintr-un element
(capul listei) s, i altă listă (coada noii liste)
– o valoare (constructor fără argumente) pentru lista vidă

Am putea defini tipul recursiv listă de elemente de tip ’a ca fiind:
type ’a list = Nil | Cons of ’a * ’a list
(un tip sumă dintre un tip cu doar o valoare, s, i un tip produs)
Cons(2, Cons(3, Nil)) reprezintă lista [2; 3]

Putem prelucra o valoare de acest tip prin potrivire de tipare:

let rec len = function
| Nil -> 0
| Cons (_, t) -> 1 + len t

În ML tipul listă e predefinit ı̂n acest fel:
constructorul de listă vidă e []
constructorul cu două argumente e scris ca operator infix ::



Lista ca tip de date abstract (TDA)

Un tip de date abstract e un tip de date definit prin operat, iile
care se pot efectua asupra lui (si constrângerile ı̂ntre acestea),
fără a expune modul ı̂n care acestea sunt implementate.

TDA izolează (abstractizează) definit, ia/interfat, a de implementare.
Ne permite implementări modificabile s, i interschimbabile, fără a
afecta programul care le foloses, te doar prin interfat, ă.

TDA listă L cu tip de element E e deobicei definit prin:
nil : ()→ L (constructorul de listă vidă)
cons : E × L→ L (constructorul pentru liste)
hd : L→ E (capul listei)
tl : L→ L (coada listei)

s, i axiomele hd(cons(e, l)) = e s, i tl(cons(e, l)) = l

Această definit, ie abstractă e suficientă pentru a defini lista
ca obiect matematic s, i a rat, iona despre proprietăt, ile listelor.



Iterarea cu funct, ii fără nume

E natural să vrem să prelucrăm fiecare element ı̂n acelas, i fel

În limbajele funct, ionale, funct, iile sunt valori, nu au nevoie de nume
pot scrie direct (fun x -> x + 5) 3

Vrem să spunem: “aplică fun x -> x + 5 pe toată lista”

Funct, ii fără nume (“lambda-expresii”) există ı̂n
λ-calcul (Alonzo Church, 1930) – baza limbajelor funct, ionale
1958: LISP (John McCarthy)
1973: ML (Robin Milner)
2007: C# v3.0
2011: C++11
2014: Java 8

Învăt, at, i concepte ! (chiar dacă adoptarea durează...)



Parcurgeri standard pentru liste

E natural să prelucrăm toate elementele unei liste.

Distingem trei cazuri principale

1. Facem ceva pentru fiecare element (ex. tipărim)
(fără a produce vreo valoare ca rezultat) List.iter

2. Transformăm fiecare element al listei cu aceeas, i funct, ie
obt, inem o nouă listă List.map

3. Combinăm toate valorile din listă List.fold_left
(le acumulăm succesiv ı̂ntr-un rezultat) List.fold_right

Acestea sunt funct, ii de iterare pentru liste (a itera = a repeta).

Scriem doar funct, ia pentru un pas de prelucrare (un element),
iar lista e parcursă automat de funct, iile standard de iterare.



Iterarea peste toate elementele listei
List.iter : (’a -> unit) -> ’a list -> unit

List.iter f [e1; e2; ... en] apelează f e1; f e2; ... f en

Funct, ia f: folosită pentru efectul (ex. tipărire), nu valoarea ei
ML are tipul unit, cu singura valoare notată () (C are void)

let rec iter f = function
| [] -> () (* orice functie are valoare, aici () *)
| h :: t -> f h; iter f t (* a; b are valoarea b *)

sau, cu o funct, ie auxiliară care evită retransmiterea parametrului f:
let iter f =

let rec iter1 = function
| [] -> ()
| h :: t -> f h; iter1 t

in iter1

List.iter print_int [1;2;3] (* tipareste 123 fara spatii *)
List.iter (Printf.printf "%d ") [1;2;3]
1 2 3 - : unit = () (* tipareste 1 2 3 cu spatii *)



Transformarea tuturor elementelor dintr-o listă
List.map : (’a -> ’b) -> ’a list -> ’b list

List.map [e1; e2; ... en] e lista [f e1; f e2; ... f en]

Rezultatul lui f poate avea alt tip decât parametrul
putem obt, ine o listă cu alt tip de elemente

let rec map f = function
| [] -> []
| h :: t -> f h :: map f t

sau, cu o funct, ie auxiliară
let map f =

let rec map1 = function
| [] -> []
| h :: t -> f h :: map1 t

in map1

[ e1 ; e2 ; ... en ]

f f ... f

[ r1 ; r2 ; ... rn ]

List.map ((+) 2) [3; 7; 4] (* lista [5; 9; 6] *)
List.map String.length ["acesta";"e";"un";"test"]
- : int list = [6; 1; 2; 4] (* lista lungimilor sirurilor *)



Combinarea elementelor dintr-o listă (de la cap)
List.fold_left : (’a -> ’b -> ’a) -> ’a -> ’b list -> ’a

List.fold_left f r0 [e1; e2; ... en] = f (...f (f r0 e1) e2...) en
[ e1 ; e2 ; ... en ]

r0 f r1 f r2 ... rn−1 f rn

let fold_left f =
let rec fold1 a = function

| [] -> a
| h :: t -> fold1 (f a h) t

in fold1

List.fold_left prelucrează elementele de la cap, e tail-recursive.
List.fold_left (+) 0 [3; 7; 4] (* suma pornind de la 0: 14 *)

List.fold_left (fun s e -> s + String.length e) 0 ["a";"si";"b"]
- : int = 4 (* suma lungimilor sirurilor din lista *)

0 + length "a"= 1→ 1 + length "si"= 3→ 3 + length "b"= 4



Cum funct, ionează s, i cum folosim List.fold_left

List.fold_left: un ciclu, odată pentru fiecare element al listei
calculează un rezultat, actualizat la fiecare iterat, ie
(pentru fiecare element)

List.fold_left are nevoie de 3 parametri:
1. o funct, ie f cu 2 parametri de tip ’a -> ’b -> ’a

p1: rezultatul calculat până acum de tip ’a
p2: elementul curent din listă de tip ’b

rezultatul f(p1, p2) devine p1 ı̂n apelul cu următorul element
2. valoarea init, ială de tip ’a

(rezultatul pentru lista vidă, s, i p1 pentru primul apel al lui f)
3. lista de prelucrat de tip ’b list

Un limbaj imperativ ar folosi o variabilă, atribuită la fiecare iterat, ie
În List.fold_left, rezultatul funct, iei f ı̂n fiecare iterat, ie e folosit
de f ı̂n următoarea iterat, ie (ca prim parametru).



Exemple de funct, ionare pentru List.fold_left

Minimul unei liste
let list_min = function

| [] -> invalid_arg "empty list" (* exceptie *)
| h :: t -> List.fold_left min h t

list_min [3; 9; -2; 4] -> List.fold_left min 3 [9; -2; 4]
(fun m e -> min m e) 3 9 -> min 3 9 = 3
(fun m e -> min m e) 3 (-2) -> min 3 (-2) = -2
(fun m e -> min m e) (-2) 4 -> min (-2) 4 = -2

Inversarea unei liste:
capul listei rămase de inversat devine capul rezultatului acumulat
let rev = List.fold_left (fun t h -> h :: t) [] [3; 7; 5]
(fun t h -> h :: t) [] 3 -> 3 :: [] = [3]
(fun t h -> h :: t) [3] 7 -> 7 :: [3] = [7; 3]
(fun t h -> h :: t) [7; 3] 5 -> 5 :: [7; 3] = [5; 7; 3]

List.rev există ca funct, ie standard pentru liste



Combinarea elementelor dintr-o listă (de la coadă)

List.fold_right : (’a -> ’b -> ’b) -> ’a list -> ’b -> ’b

fold_right f [e1; e2;...; en] rn = f e1 (f e2 (...(f en rn)...))
fold_right calculează, de la dreapta la stânga rezultatul
r0 = f e1 r1 ← r1 = f e2 r2 ← ... rn−1 = f en rn

[ e1 ; e2 ; ... en ]

r0 f r1 f r2 ... rn−1 f rn

let fold_right f lst b =
let rec foldf b = function

| [] -> b
| h :: t -> f h (foldf b t)

in foldf b lst

fold_right prelucrează elementele de la coadă, nu e tail-recursive.



Exemplu: filtrarea unei liste

List.filter : (’a -> bool) -> ’a list -> ’a list

List.filter f [a1; a2; ...; an] :
lista elementelor ak pentru care f ak e adevărată

let filter f = (* f e fixata pentru functia interioara filt1 *)
let rec filt1 = function (* doar un parametru lista *)

| [] -> [] (* din lista vida nu avem ce selecta *)
| h :: t -> let ft = filt1 t in (* selecteaza din coada t *)

if f h then h :: ft else ft (* ia h daca e bun *)
in filt1 (* adica: let filter f lst = filt1 lst *)

List.filter (fun x -> x mod 3 = 0) [1;2;3;4;5;6];;
- : int list = [3; 6]
List.filter ((<) 3) [1;2;3;4;5;6];; (* fun x -> 3 < x *)
- : int list = [4; 5; 6] (* (<) 3 = fun x -> (<) 3 x *)

Exercit, iu: scriet, i filter folosind List.fold_right



Exemplu: ciurul lui Eratostene

(* lista numerelor de la a la b *)
let fromto a =

let rec from2 r b = if b<a then r else from2 (b::r) (b-1)
in from2 []

let nondiv d x = x mod d <> 0

let rec sieve = function
| [] -> []
| h :: t -> h :: sieve (List.filter (nondiv h) t)

let primes = sieve (fromto 2 10000)



Important, a funct, iilor de parcurgere

Separă partea mecanică de cea funct, ională
(parcurgerea standard a listei de prelucrarea specifică problemei)

Nu necesită scrierea (repetată) a codului de parcurgere.
Intent, ia prelucrării poate fi scrisă mai clar (mai direct).
Reduce probabilitatea erorilor la sfârs, itul prelucrării (lista vidă)
În multe cazuri, această parcurgere standard poate fi paralelizată

Aceste idei au fost preluate dincolo de limbajele funct, ionale
Java 8 introduce interfat, a Stream<T>:

are metode de iterare similare (map, filter, reduce)
permite paralelizarea lor
permite prelucrări cu funct, ii anonime (lambda expressions)



De s, tiut

Lucrul cu liste prin potrivire de tipare

Lucrul cu funct, ii standard de parcurgere

Prelucrări care au ca parametri funct, ii


